

# NAG Library Routine Document

## D01AHF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

D01AHF computes a definite integral over a finite range to a specified relative accuracy using a method described by Patterson.

### 2 Specification

```
FUNCTION D01AHF (A, B, EPSR, NPTS, RELERR, F, NLIMIT, IFAIL)
REAL (KIND=nag_wp) D01AHF
INTEGER           NPTS, NLIMIT, IFAIL
REAL (KIND=nag_wp) A, B, EPSR, RELERR, F
EXTERNAL         F
```

### 3 Description

D01AHF computes a definite integral of the form

$$\int_a^b f(x) dx.$$

The method uses as its basis a family of interlacing high precision rules (see Patterson (1968)) using 1, 3, 7, 15, 31, 63, 127 and 255 nodes. Initially the family is applied in sequence to the integrand. When two successive rules differ relatively by less than the required relative accuracy, the last rule used is taken as the value of the integral and the operation is regarded as successful. If all rules in the family have been applied unsuccessfully, subdivision is invoked. The subdivision strategy is as follows. The interval under scrutiny is divided into two sub-intervals (not always equal). The basic family is then applied to the first sub-interval. If the required accuracy is not obtained, the interval is stored for future examination (see IFAIL = 2) and the second sub-interval is examined. Should the basic family again be unsuccessful, then the sub-interval is further subdivided and the whole process repeated. Successful integrations are accumulated as the partial value of the integral. When all possible successful integrations have been completed, those previously unsuccessful sub-intervals placed in store are examined.

A large number of refinements are incorporated to improve the performance. Some of these are:

- (a) The rate of convergence of the basic family is monitored and used to make a decision to abort and subdivide before the full sequence has been applied.
- (b) The  $\epsilon$ -algorithm is applied to the basic results in an attempt to increase the convergence rate. See Wynn (1956).
- (c) An attempt is made to detect sharp end point peaks and singularities in each sub-interval and to apply appropriate transformations to smooth the integrand. This consideration is also used to select interval sizes in the subdivision process.
- (d) The relative accuracy sought in each sub-interval is adjusted in accordance with its likely contribution to the total integral.
- (e) Random transformations of the integrand are applied to improve reliability in some instances.

## 4 References

Patterson T N L (1968) The Optimum addition of points to quadrature formulae *Math. Comput.* **22** 847–856

Wynn P (1956) On a device for computing the  $e_m(S_n)$  transformation *Math. Tables Aids Comput.* **10** 91–96

## 5 Arguments

- 1: A – REAL (KIND=nag\_wp) *Input*  
*On entry:*  $a$ , the lower limit of integration.
- 2: B – REAL (KIND=nag\_wp) *Input*  
*On entry:*  $b$ , the upper limit of integration. It is not necessary that  $a < b$ .
- 3: EPSR – REAL (KIND=nag\_wp) *Input*  
*On entry:* the relative accuracy required.  
*Constraint:* EPSR > 0.0.
- 4: NPTS – INTEGER *Output*  
*On exit:* the number of function evaluations used in the calculation of the integral.
- 5: RELERR – REAL (KIND=nag\_wp) *Output*  
*On exit:* a rough estimate of the relative error achieved.
- 6: F – REAL (KIND=nag\_wp) FUNCTION, supplied by the user. *External Procedure*  
F must return the value of the integrand  $f$  at a given point.

The specification of F is:

```
FUNCTION F (X)
REAL (KIND=nag_wp) F
REAL (KIND=nag_wp) X
```

1: X – REAL (KIND=nag\_wp) *Input*

*On entry:* the point at which the integrand  $f$  must be evaluated.

F must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub) program from which D01AHF is called. Arguments denoted as *Input* must **not** be changed by this procedure.

- 7: NLIMIT – INTEGER *Input*  
*On entry:* a limit to the number of function evaluations. If NLIMIT  $\leq 0$ , the routine uses a default limit of 10000.
- 8: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output

arguments may be useful even if  $IFAIL \neq 0$  on exit, the recommended value is  $-1$ . **When the value  $-1$  or  $1$  is used it is essential to test the value of  $IFAIL$  on exit.**

*On exit:*  $IFAIL = 0$  unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry  $IFAIL = 0$  or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

**Note:** D01AHF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

$IFAIL = 1$

The integral has not converged to the accuracy requested. It may be worthwhile to try increasing NLIMIT.

$IFAIL = 2$

Too many unsuccessful levels of subdivision have been invoked.

$IFAIL = 3$

On entry,  $EPSR \leq 0.0$ .

$IFAIL = -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

$IFAIL = -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

$IFAIL = -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

When  $IFAIL = 1$  or  $2$  a result may be obtained by continuing without further subdivision, but this is likely to be **inaccurate**.

## 7 Accuracy

The relative accuracy required is specified by you in the variable EPSR. The routine will terminate whenever the relative accuracy specified by EPSR is judged to have been reached.

If on exit,  $IFAIL = 0$ , then it is most likely that the result is correct to the specified accuracy. If, on exit,  $IFAIL = 1$  or  $2$ , then it is likely that the specified accuracy has not been reached.

RELERR is a rough estimate of the relative error achieved. It is a by-product of the computation and is not used to effect the termination of the routine. The outcome of the integration must be judged by the value of  $IFAIL$ .

## 8 Parallelism and Performance

D01AHF is not threaded in any implementation.

## 9 Further Comments

The time taken by D01AHF depends on the complexity of the integrand and the accuracy required.

## 10 Example

This example evaluates the integral to a requested relative accuracy of  $10^{-5}$

$$\int_0^1 \frac{4}{1+x^2} dx = \pi.$$

### 10.1 Program Text

```
! D01AHF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module d01ahfe_mod

! D01AHF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp
! .. Implicit None Statement ..
Implicit None
! .. Accessibility Statements ..
Private
Public                                :: f
! .. Parameters ..
Integer, Parameter, Public           :: nin = 5, nout = 6
Contains
Function f(x)

! .. Function Return Value ..
Real (Kind=nag_wp)                   :: f
! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: x
! .. Executable Statements ..
f = 4.0E0_nag_wp/(1.0E0_nag_wp+x*x)

Return

End Function f
End Module d01ahfe_mod
Program d01ahfe

! D01AHF Example Main Program

! .. Use Statements ..
Use nag_library, Only: d01ahf, nag_wp
Use d01ahfe_mod, Only: f, nin, nout
! .. Implicit None Statement ..
Implicit None
! .. Local Scalars ..
Real (Kind=nag_wp)                   :: a, ans, b, epsr, relerr
Integer                                :: ifail, nlimit, npts
! .. Executable Statements ..
Write (nout,*) 'D01AHF Example Program Results'

Read (nin,*)
Read (nin,*) a, b
Read (nin,*) nlimit
Read (nin,*) epsr

ifail = -1
ans = d01ahf(a,b,epsr,npts,relerr,f,nlimit,ifail)

Select Case (ifail)
```

```
Case (0:2)
  Write (nout,*)
  Write (nout,99999) 'Integral = ', ans
  Write (nout,*)
  Write (nout,99998) 'Estimated relative error = ', relerr
  Write (nout,*)
  Write (nout,99997) 'Number of function evaluations = ', npts
End Select

99999 Format (1X,A,F8.5)
99998 Format (1X,A,E10.2)
99997 Format (1X,A,I5)
End Program d01ahfe
```

## 10.2 Program Data

```
D01AHF Example Program Data
  0.0    1.0      : a, b
  0      : nlimit
  0.00001 : epsr
```

## 10.3 Program Results

```
D01AHF Example Program Results
```

```
Integral = 3.14159
```

```
Estimated relative error = 0.58E-08
```

```
Number of function evaluations = 15
```

---